





=====When_to_sample?=====



The idea is to sample when something happens.
So when is something happening?



Threshold crossing might lose information when the input signal becomes small.
There is no reason why the sample thresholds need to be uniformly spaced voltage wise.



This example triggers additionally off of the waveform's **minimums** and **maximums**.
Spice uses a type of sampling that look like it responds more to waveform acceleration.




For the last three waveform cycles above,
notices that only zero crossings and max and min data points are sampled.
That averages to about **four crossing points per cycle**.
And still the extrapolated curve in green comes close to the real input in red.
So the trade off between number of sample points and signal integrity is not obvious.

Remember for Asynchronous Analog to Digital Conversion,
the **sampling is done only when something is happening**.
And the input signal is sampled at **a precise value**,
at **a precise time**.
So how well does it capture a waveform?

====Spice_Code=====

Extrapolate_Decaying_Sinewave



```

*====Create Signal=====
*V_SIN#  NODE_P  NODE_N  DC  VALUE  SIN(  V_DC  AC_MAG  FREQ  DELAY  FDamp)
VIN      VSIN    0      DC    0      SIN(    0      .5      50      )
Vtime    Vtime    0      DC    0      PWL( 0 0 1 1 )
Bgain    Vgain    0      V =   exp(-V(Vtime)/.02)
BVIN     VIN     0      V =   V(VSIN)*V(Vgain) +.5
    
```

```

*====AADC=====
BTH0     D0      0      V =   u( V(VIN)  -1/2)
BOFF0    VIN    OFF0    V =   V(D0)/2
BTH1     D1      0      V =   u( V(OFF0) -1/4)
BOFF1    OFF0  OFF1    V =   V(D1)/4
BTH2     D2      0      V =   u( V(OFF1) -1/8)
BOFF2    OFF1  OFF2    V =   V(D2)/8
BTH3     D3      0      V =   u( V(OFF2) -1/16)
BOFF3    OFF2  OFF3    V =   V(D3)/16
BTH4     D4      0      V =   u( V(OFF3) -1/32)
BDAC     VOUTA  0      V =   (V(D0)/2+V(D1)/4+V(D2)/8 +V(D3)/16)
    
```

```

*====Counter_Trigger=====
CHP      VOUTA  HP      .1n
RHP      HP      0      100k
CHP2     VIN    HP2     .1n
RHP2     HP2    0      100k
BTRIG2   trig2  0      V = u( 100*abs(V(HP))  -.1  )+u(.005m - abs(V(hp2)) )
    
```

```

*====Counter=====
Btrig    trig  0      V =   1*u( V(VOUTA)- V(VIN) +5m )
XTOGLE   trig2 D0B    TOGGLE
XTOGLE0  D0B   D1B    TOGGLE
XTOGLE1  D1B   D2B    TOGGLE
XTOGLE2  D2B   D3B    TOGGLE
XTOGLE3  D3B   D4B    TOGGLE
XTOGLE4  D4B   D5B    TOGGLE
BDAC2    VOUTB 0      V =   63-(V(D5B)*32+V(D4B)*16+V(D3B)*8+V(D2B)*4 +V(D1B)*2 +V(D0B))
    
```

```

XPOS E2   trig2  PEOUT  POS E2
XS_H1     Vtime  trig    VtimeS  SH
    
```

```

.control
*TRAN    TSTEP  TSTOP  TSTART  TMAX  ?UIC?
tran     2u    100m   0        2u
*====Value And Time=====
set      pensize = 2
let      numb = length(vtimes)-1001
unlet    index
unlet    val
unlet    tref
let index = vector(43)*0 +42
let val   = vector(43)*0 +.5
let tref  = vector(43)*0 +.1
let dt    = vector(43)*0
plot      vin  vouta  voutb/50 xlimit 1m 100m
    
```

```

*====Reduce_Number_Of_Points=====
let i    = 1000
repeat   $&numb
let tt   = peout[i]

if       (tt > .1)
let k    = voutb[i]
let      index[k] = k
let      value = vin[i]
let      val[k] = value
let      tref[k] = time[i]
end
let i    = i+1
end

let i    = 1
repeat   41
let dt[i] = tref[i] -tref[i-1]
let i    = i+1
end
    
```

```

plot val  tref*10  vs index
plot val          vs tref pointplot
    
```

*====Curvefit_Points=====

```

unlet x
unlet indext
unlet indexn
let x = vector(430)*0 +.1
let indext = vector(430)
let indexn = vector(430)*0
let y = vector(430)*0 +.5
let z = vector(430)*0 +.5

let i = 0
let j = 1
let n = 0
repeat 40
let S0 = val[j ]-val[j-1]
let S1 = val[j+1]-val[j ]
let S2 = val[j+2]-val[j+1]
let S01 = (S1+S0)/2
let S12 = (S2+S1)/2
let HH = val[j+1]-val[j]
let A = val[j]
let B = S01
let C = 3*HH -2*S01 -S12
let D = -2*HH +S01 +S12

repeat 10
let indext[i] = i
let x[i] = tref[j] + n*(tref[j+1]-tref[j])/10
let tt = n/10
let indexn[i] = n/10
let y[i] = A +B*tt +C*tt^2 +D*tt^3
let z[i] = .5*sin( 360*50*x[i] )*exp(-x[i]/.02) +.5
let n = n + 1
let i = i + 1
end
let n = 0
let j = j + 1
end

```

```

plot z vs x
plot z y vs x

```


.endc

```

.MODEL SW SW( VT=.5 VH=.1 RON=1 ROFF=100MEG)
.MODEL DD2 D( IS=3.15e-18 RS=1 CJO=2f)

```

*=====POS_Edge2=====




```

.SUBCKT POS_E2 IN OUT
BBUF VBF 0 V = 2*u( v(IN )-.5 )
RLP VBF VLP 10k
CLP VLP 0 .1n IC=0
BAND OUT 0 V = u( u(v(VBF )-.5)*u(.5 -v(VLP ) ) -.1)
.ENDS POS_E2

```

*=====Sample_Hold=====




```

.SUBCKT SH IN CNTL OUT
B1 IN2 0 V = v( IN )
S1 IN2 IN3 CNTL 0 SW
C1 IN3 0 .1u
R1 IN3 0 100Meg
B2 OUT 0 V = v( IN3 )
.ENDS SH

```

*=====POS_Edge=====



```

.SUBCKT POS_E IN OUT
BBUF VBF 0 V = u( v(IN )-.5 )


```

```

RLP      VBF      VLP      10k
CLP      VLP      0         .1n      IC=0
BAND     OUT      0         V =      u( u(v(VBF )-.5)*u(.5 -v(VLP ) ) -.1)
.ENDS    POS_E

```

-----TOGGLE-----



```

.SUBCKT  TOGGLE  IN      OUT2
BBUF     VBF      0       V =      u( v(IN )-.5 )
RLP      VBF      VLP     1k
CLP      VLP      0       5n
BAND     VPE      0       V =      u( u(v(VBF )-.5)*u(.5 -v(VLP ) ) -.5)
RLP2     VPE      VLP2    .3k
CLP2     VLP2     0       2n
BBRK     BRK      0       V =      1-u(v(VLP2 )-.2)
EMAK     MAK      0       V =      u(v(VLP2 )-.9)
S1       OUT      VC1     MAK      0       SW
S2       NVC1     VC2     BRK      0       SW
C1       VC1      0       10n
C2       VC2      0       10n
R1       VC1      0       10Meg
R2       VC2      0       10Meg
BINV     NVC1     0       V =      1-u(v(VC1)-.2)
BOUT     OUT      0       V =      u(v(VC2)-.2)
BOUT2    OUT2     0       V =      1- u(v(VC2)-.2)
.ENDS    TOGGLE

```

.end

8.18.11_4.09PM
dsauersanjose@aol.com
Don Sauer